Appendix for "Measuring the impact of Organizational Intervention on Absence of Employees with Disabilities: A Quasi-Experimental Design"

Supplemental Online material

Technical documentation of the article "Measuring the impact of Organizational Intervention on Absence of Employees with Disabilities: A Quasi-Experimental Design" submitted to *International Labour Review*.

Description of the dataset

Number of episodes of absence



Figure 1: Distribution of the number of yearly episodes of absence by individual.

Model results

Baseline latent rate of absence



Figure 2: Caterpillar plot with median (dot), 90 and 95 percent credible intervals/Highest Posterior Densities (thick and thin lines, respectively) of the distribution of γ, the parameter that accounts for the baseline rate of absence. Therefore, a value of -0.8 (illness) implies that the overall expected rate of illness is $\exp(-0.8) = 0.45$, or almost half an absence due to illness per worker. For injuries it is $\exp(-1.85) = 0.16$ expected episodes of absence every year due to injury, or a bit less than one absence due to injury every 5 years.

Temporal rate of absence



Figure 3: Caterpillar plot with median (dot), 90 and 95 percent credible intervals (thick and thin lines, respectively) of the distribution of κ, the parameter that accounts for the overall yearly rates of absence. It captures the general effects of each calendar year on absence. Not surprisingly, injuries are stable over time, whereas illness has a higher variation, which may be due to prevalence of diseases in specific years, or to general economic conditions.

Profile-specific rate of absence



Figure 4: Caterpillar plot with median (dot), 90 and 95 percent credible intervals (thick and thin lines, respectively) of the distribution of δ, the parameter that accounts for the profile differences in prevalence of absence for the entire time period considered.

Effects of the lagged outcome variable



Figure 5: Caterpillar plot with median (dot), 90 and 95 percent credible intervals (thick and thin lines, respectively) of the distribution of ϕ, the parameter that accounts for the effect of the lagged outcome variable.

Individual-specific effects on absence



Figure 6: Caterpillar plot with median (dot), 90 and 95 percent credible intervals (thick and thin lines, respectively) of the distribution of β, the parameter that accounts for the effect of the individual-specific variables. A value of -1 (pre-retirement in injuries) can be interpreted as follows: Compared to the reference individual (fixed contract), workers with temporary contracts are $\exp(-1) = 0.36$ times less likely to have injuries, or $1 - 0.36 = 64$ percent less. Or, for individuals with mental disability, they are $\exp(0.5) = 1.64$ more likely to have an annual episode of absence due to injuries (64 percent more).

Model specification in JAGS/BUGS

```
model {
  for (n in 1:nN) {
    y.n.absence[n] ~ dnegbin(p[n], r[type[n]])
    p[n] <- r[type[n]]/ (r[type[n]] + lambda[n])
    lambda[n] <- exp(Mu[n])
    Mu[n] <-  phi[type[n]] * y.n.absence.lag[n]
           + mu[profile[n], year.t[n], type[n]]
           + beta[1, type[n]] * years.firm[n]
           + beta[2, type[n]] * female[n]
           + beta[3, type[n]] * age[n]
           + beta[4, type[n]] * head.brigade[n]
           + beta[5, type[n]] * temporary.contract[n]
           + beta[6, type[n]] * pre.retirement.contract[n]
           + beta[7, type[n]] * disability.high[n]
           + beta[8, type[n]] * disability.type.physical.sensorial[n]
           + beta[9, type[n]] * disability.type.mental[n]
  }
  for (tp in 1:nTP) {
    phi[tp] ~ dnorm(0, 0.001)
    r[tp] ~ dunif(0, 10)
    for (b in 1:9) {
      beta[b,tp] ~ dnorm(0, 0.01)
    }
    for (y in 1:nY) {
      for (p in 1:nProfiles) {
        mu[p,y,tp] <- gamma[tp]        # type of absence latent rate for all years
                    + kappa[y,tp]      # anual variation, by type of absence
                    + delta[p,tp]      # profile differences on the latent rate
                    + theta[p,(years.from.intervention[y]+1),tp]
                                       # theta: intervention effect
      }
    }
  }
  # gamma
  for (tp in 1:nTP) {
    gamma[tp] ~ dnorm(0, tau.gamma[tp])
    tau.gamma[tp] <- pow(sigma.gamma[tp], -2)
    sigma.gamma[tp] ~ dunif(0, 20)
  }
  # kappa
  for (tp in 1:nTP) {
    for (y in 1:nY) {
      kappa[y,tp] ~ dnorm(0, tau.kappa[tp])
    }
    tau.kappa[tp] <- pow(sigma.kappa[tp], -2)
    sigma.kappa[tp] ~ dunif(0, 10)
  }
  # delta
  for (tp in 1:nTP) {
    # EnD
    delta[1,tp] <- 0
    # EWD non participating
    delta[2,tp] ~ dnorm(0, tau.delta[2])
    ## EWD participating
    delta[3,tp] ~ dnorm(0, tau.delta[3])
  }
  # sigma_delta
  for (p in 1:nProfiles) {
    tau.delta[p] <- pow(sigma.delta[p], -2)
    sigma.delta[p] ~ dunif(0, 10)
  }
  # theta
  for (tp in 1:nTP) {
    for (y in 1:(nY.from.intervention + 1)) {
      theta[1,y,tp] <- 0 # EnD
    }
    theta[2,1,tp] <- 0 # before intervention
    theta[3,1,tp] <- 0 # before intervention
    for (y in 2:(nY.from.intervention + 1)) {
      # EWD:
      theta[2,y,tp] ~ dnorm(0, 1/(0.5^2))
      # EWD participating:
      theta[3,y,tp] ~ dnorm(log(0.5), 1/(0.5^2))
    }
    for (t in 1:2) {
      tau.theta[t,tp] <- pow(sigma.theta[t,tp], -2)
      sigma.theta[t,tp] ~ dunif(0, 10)
    }
  }
}
```